Controlling Stepper Motor

MDA-8086 Kit – PPI Application

Birzeit University

Information Technology Faculty

Computer Systems Engineering Department

Abstract

This experiment aims at understanding and testing the main concepts of stepper motor operation using the 82C55 Programmable Peripheral Interface (PPI) to control or positioning Stepper Motor.

The MDA 8086 subsystem will be used to control and position the Stepper Motor.

PART I Theoretical and Technical Introduction

1.1 Introduction to Stepper Motors

The stepping motor is a device which can transfer the incoming pulses to stepping motion of a predetermined angular displacement. By using suitable control circuitry the angular displacement can be made proportional to the number of pulses. Using microcomputer, one can have better control of the angular displacement resolution and angular speed of a stepping motor. In the past few years the stepping motor has improved in size reduction, speed and precision. Stepping motor already have and will continue to have wide applications in industry.

Stepping motors are suitable for translating digital inputs into mechanical motion. In general, there are three types of stepping motor (For detailed Comparison look at Table 1):

1. VR (Variable Reluctance) stepping motors
2. Hybrid Stepping motors
3. PM (Permanent Magnet) Stepping motors

Table 1 : Stepping Motors Characteristics Comparison

Motor Type Characteristics	PM	VR	Hybrid
Efficiency	High	Low	High
Rotor Inertia	High	Low	Low
Speed	High	High	Low
Torque	Fair	Low	High
Power O/P	High	Low	Low
Damping	Good	Poor	Poor
Typical Step Angle	1.8°, 15°, 30°	7.5°, 15°, 30°	0.18°, 0.45°

A stepper motor has several coils arranged in a circle, and the rotating shaft is inside the circle. Whenever a coil is excited, the shaft is attracted towards the coil and rotates to point to the coil. Therefore, if there are n coils, there will be n positions that the shaft can rotate to, depending on which coil is excited. If two adjacent coils are excited, the shaft will rotate to in between the two coils. So there are n more positions that the shaft can rotate to. Altogether, if there are n coils, there are 2n positions.

One-Phase Excitation is when only one coil is excited at a time, so that the shaft rotates from one coil to another. **Two-Phase Excitation** is when two adjacent coils are excited together, so that the coil rotates to positions between the coils. **1-2 Phase Excitation** is when the coils are excite in such a manner that the shaft rotates from a coil to between two coils and vice versa (it uses both 1-phase and 2-phase excitation).

1.2 Single-phase excitation

Figure 1 is used to explain the operation of simplified stepping motor (90°/step). Here the **A** coil and **B** coil are perpendicular to each other. If either **A** or **B** coil is excited (a condition which is known as single-phase excitation), the rotor can be moved to 0°, 90°, 180°, 270° degree position depending on the current's ON/OFF conditions in the coils . Table 1 clarifies more the stepping motor operation.



Figure 1 Bipolar Stepper motor Coil schematic

Table 2 Stepper Motor Clockwise Rotation One-Phase

Step	Coil 4	Coil 3	Coil 2	Coil 1	
a.1	**On**	off	Off	off	
a.2	Off	**on**	off	off	

a.3	Off	off	**on**	off	
a.4	Off	off	off	**on**	

From Table 2 we can derive stepper motor excitation values. Table 2 shows excitation **values** for stepper motor with starting position at coil A (clockwise). Fill Table3 (Pre-lab)

Table 3 : Excitation table of One-Phase excitation (clockwise)

	A	B	A'	B'	HEX
Step1	0	1	1	1	07H
Step2	1	0	1	1	0BH
Step3	1	1	0	1	0DH
Step4	1	1	1	0	0EH

1.3 Two–Phase Excitation

Two-Phase excitation is when two adjacent coils are excited together, so that the coil rotates to positions between the coils. In **Two-Phase** excitation, both coils have current flowing at the same time, then the rotor positions can be 45°, 135°, 225°, 315° degrees as shown in Figure 3. This is known as two-phase exception.

In 2-1 Phase excitation , the excitation alternates between 1-phase and 2-phase, then the motor will rotate according to 0°, 45°, 90°, 135°, 180°, 225°, 270°, 315° sequence. This is 1-2 phase excitations; each step distance is only half of step movement of either One-Phase or Two-Phase excitation.

Figure 2 illustrates clockwise of Two-Phase excitation.



Figure 2 Clockwise Two-Phase Excitation

Commercial stepping motor uses multi-motor rotor, the rotor features two gear like PM cylinders that are turned one-half of tooth spacing. One gear is South Pole; the other gear is North Pole. If a 50-tooth rotor gear is used, the following movement sequence will proceed.

A. Single-phase excitation:

   The stepping position will be 0°, 1.8°, 3.6°, …, 358.2°, total 200 steps in one round.

B. Two-phase excitation:

   The stepping position will be 0.9°, 2.7°, 4.5°, …, 359.1°, total 200 steps in one round.

C. Single-phase and two-phase excitations combined:

   The stepping position will be 0°, 0.9°, 1.8°, 2.7°, 3.6°, 4.5°, …, 358.2°, 359.1°, total 400 steps in one round.

Since stepping motor makes step-by-step movement and each step is equidistant, the rotor and stator magnetic fields must be synchronous. During start-up and stopping, the two fields may not be synchronous, so it is suggested to slowly accelerate and decelerate the stepping motor during the start-up or stopping period.

1.4 Industrial Applications

1.4.1 Robotic Arm

In a motor-car assembly industry, robotic arms are constructed with stepper motors. We can use 3 motors to move the arm along 3 axes. Programs are written with particular number of steps and the time for turning for each direction movement stored in the memory. Time delay has to be accurately calculated, otherwise, after a long time has passed, the arm will lag behind the assemblage line by a large amount of time.

1.4.2 Satellite or Solar Tracking

A satellite or solar tracking device may be made with stepper motors moving the shafts of an antenna or a solar panel. According to measured signal or light intensity, the microcomputer will decide if to turn the panel/antenna in a certain direction, and if so, by how many steps.

1.5 Interfacing and Isolating Circuitry

The 8255 IC U29 on the MDA-8086 is connected via its port B upper nibble to the stepper motor interface. The connection is done using pins 22 to 25 of the P3 connector shown in Figure 3, which in turn are connected to pins 22 to 25 of P4 connector shown in Figure 4.



Figure 3 The 8255 Interface with Stepper Motor on MDA-8086

As long as stepper motor (or any other kind of motor) draws a huge amount of current from the source, we cannot directly connect a stepper motor to the digital IC's, which have source and sink current limitation. Therefore, we use the logic output of the digital interface to drive a transistor- or MOS-driven motor driver, which are capable of supplying enough current to drive a motor. Careful connections should be made with consideration of the current ratings of the components. The connections shown in Figure 4 are for unipolar-type stepper motors.



Figure 4 Transistor-drive for stepper motor on board

The P10 connector in Figure 4 provides the interface between the stepper motor drivers that are on the MDA board and the stepper motor itself. The P6 6-pins are connected to the motor wires as shown in Figure 5:



Figure 5 The P6 6-pins are connected to the motor wires

The coil schematic of a bipolar stepper motor is shown in Figure 6:



Figure 6 Bipolar Stepper motor Coil schematic

Sending 1011 to the $A\bar{A}B\bar{B}$ pins will rotate the motor to a particular location, by energizing Coil A in a particular direction. Then rotating this sequence to 0111, will energize Coil B in the same direction. Motor will turn by another step. Then, another rotated sequence of 1110 will energize the Coil A with a reversed direction as compared to the previous one, so the motor will continue with another step in the same direction. Then 1101 will energize the Coil B in the reverse direction, and one full rotation will be completed, if the rotor has only two teeth.

However, commercial motors have more than two teeth, and they turn by steps of typically 1.8° instead of 90°.

PART II Pre-Lab

(This part should be handed on to the teaching assistant in your Lab)

1. Review of PPI (82C55) Chip Datasheet and understand its modes of operation.
2. Study the concepts of stepper motor operation.

3. What is the effect of assembly instructions ROR, ROL?
4. What does “stepper motor resolution” means?
5. Fill the 1-phase excitation table (counter-clockwise), 2-phase excitation table (clockwise).

	A	B	A’	B’	HEX
Step1					
Step2					
Step3					
Step4					

	A	B	A’	B’	HEX value
Step1					
Step2					
Step3					
Step4					

PART III Practices

3.1 PRACTICE I: C Code Controlling and Positioning Stepper Motor

Step1: Connect the external interface to the MDA-8086 board, turn it on from the external power supply and execute the following programs.

Step2: Write the following code and save it.

```c
#include "mde8086.h"
void wait( long del )
{
    while( del-- );
}

/* Output Pulse Table to the Motor. */
unsigned char step_tbl[4] = {0xe0, 0xd0, 0xb0, 0x70};

void main( void )
{
    unsigned char step;

    outportb( PPI1_CR, 0x80 );
    outportb( PPI1_B, 0xff );
    outportb( PPI1_A, 0xff );
    outportb( PPI1_C, 0X20 );

    step = 0;
    do  {
        outportb( PPI1_B, step_tbl[step++] );
        if( step >= 4 )  step = 0;
        wait( 100 );
    } while( 1);
}
```

Figure 7 Code 1

Step3: Compile and build this C file and execute it on MDA-8086 kit. (How? Review Exp#1 Intro. To MDA Kit)

TASKS:

1. Explain what does this code do?
2. In which excitation phase does the motor operate?
3. Modify the code to change the rotation direction of motor

4. Make changes to the above program for 2-phase excitation. Execute the program.

3.2 PRACTICE II: ASM Code Controlling and Positioning Stepper Motor

Step1: Write the following code and save it.

```
;PROGRAM TO RUN A UNIPOLAR STEPPER MOTOR
;WITHOUT CONTROLLING BY EXTERNAL MEANS

CODE    SEGMENT
    ASSUME  CS:CODE,DS:CODE,ES:CODE,SS:CODE
    ;
PPIC_C  EQU 1FH
PPIC    EQU 1DH
PPIB    EQU 1BH
PPIA    EQU 19H
    ;
    ORG 1000H
    ;
    ;
    MOV AL,10000000B
    OUT PPIC_C,AL
    ;
    MOV AL,11111111B
    OUT PPIA,AL
    MOV AL,00000000B
    OUT PPIC,AL


L2:     MOV     AL,11101110B

L3:     OUT     PPIB,AL

        ROL AL,1

        CALL    TIMER
        JMP     L3
    ;
TIMER:  MOV     CX,3

TIMER2: PUSH    CX
        MOV     CX,0ffh; CHANGE CX VALUE TO ADJUST SPEED
TIMER1: NOP
    NOP
    NOP
    NOP
    LOOP    TIMER1
    POP CX
    LOOP    TIMER2
    RET
;
CODE    ENDS
```

Figure 8 Code 2

Step2: Compile and build this C file and execute it on MDA-8086 kit. (How? Review Exp#1 Intro. To MDA Kit)

TASKS:

1. Explain what does this code do?
2. In which excitation phase does the motor operate?
3. Modify the code let the stepper motor rotate in opposite direction (clockwise)

3.3 PRACTICE III: ASM Code Controlling and Positioning Stepper Motor (Another Way)

Step1: Write the following code and save it.

```

;ANOTHER WAY TO IMPLEMENT THE PROGRAM
;BY ACCESSING DATA TABLE
;WE CAN PREDEFINE THE MOVEMENT OF SOME ROTATIONS
;IN THE DATA TABLE
;THIS IS USEFUL FOR PERFORMING PREDEFINED MOVEMENTS
;IN AN INDUSTRY
;CONTINUOUS RUNNING IN ONE DIRECTION
;MAY NOT BE OUR DESIRE

CODE    SEGMENT
    ASSUME  CS:CODE,DS:CODE,ES:CODE,SS:CODE
    ;
PPIC_C  EQU 1FH
PPIC    EQU 1DH
PPIB    EQU 1BH
PPIA    EQU 19H

    ;
    ORG 1000H
    ;
    ;
    MOV AL,10000000B
    OUT PPIC_C,AL
    ;
    MOV AL,11111111B
    OUT PPIA,AL
    MOV AL,00000000B
    OUT PPIC,AL
```

```
L1:     MOV     SI,OFFSET ROT
L2:     MOV     AL,BYTE PTR CS:[SI]
        CMP     AL,0FFH
        JZ      L1
        OUT     PPIB,AL
        INC     SI
        CALL    TIMER
        JMP     L2
    ;
TIMER:  MOV CX,1
TIMER2: PUSH    CX
        MOV CX,0
TIMER1: NOP
    NOP
    NOP
    NOP
    LOOP    TIMER1
    POP CX
    LOOP    TIMER2
    RET

ROT:    DB      11101110B
        DB      10111011B
        DB      11011101B
        DB      01110111B
        DB      0FFH
    ;
CODE    ENDS
    END
```

Figure 9 Code 3

Step2: Compile and build this C file and execute it on MDA-8086 kit. (How? Review Exp#1 Intro. To MDA Kit)

TASKS:

1. Explain what does this code do?
2. Modify the last code; let the stepper motor rotate in opposite direction (2-phase excitation).

3.4 PRACTICE IV: ASM Code Controlling the Speed of the Stepper Motor

In this practice you will use the kit keypad to control the stepper motor speed. To achieve this refer to Appendix A which explain how to accomplish keyboard interface.

Step1: Write the following code and save it.

```

;PROGRAM FOR CONTROLLING A STEPPER MOTOR
;DECREASE THE SPEED OF THE STEPPER MOTOR
;BY KEEPING ANY LOWER NUMBER KEY ON THE KEYPAD
;INCREASE THE SPEED OF THE STEPPER MOTOR
;BY KEEPING ANY HIGHER NUMBER KEY ON THE KEYPAD
;CHANGE THE CX VALUES TO ADJUST YOUR OWN SPEEDS

CODE    SEGMENT
    ASSUME  CS:CODE,DS:CODE,ES:CODE,SS:CODE
    ;
PPIC_C  EQU 1FH
PPIC    EQU 1DH
PPIB    EQU 1BH
PPIA    EQU 19H
KEY     EQU     01H
    ;
    ORG 1000H
    ;
    MOV AL,10000000B
    OUT PPIC_C,AL
    ;
    MOV AL,11111111B
    OUT PPIA,AL
    MOV AL,00000000B
    OUT PPIC,AL

   MOV     AL,80H
   OUT     KEY,AL
```

```
L2:     MOV     AL,11101110B
L3:     OUT     PPIB,AL

        ROL AL,1

        PUSH    AX
        IN      AL,KEY

        CMP     AL,80H
        JE      NONE

        CMP     AL,40H
        JE      ZERO



NONE:   MOV     CX,400D
        CALL    TIMER
        POP     AX
        JMP     L3
ZERO:   MOV     CX,200D
        CALL    TIMER
        POP     AX
        JMP     L3
    ;



TIMER:
TIMER2: PUSH     CX
        MOV      CX,00ffh
TIMER1: NOP
    NOP
    NOP
    NOP
    LOOP    TIMER1
    POP CX
    LOOP    TIMER2

    RET

    ;
CODE    ENDS
    END
```

Step2: Compile and build this C file and execute it on MDA-8086 kit. (How? Review Exp#1 Intro. To MDA Kit)

TASKS:

1. Explain what does this code do?
2. Modify the code to control the speed using keypads (From 0 to 9) (i.e. decrease the speed of the stepper motor by keeping any lower number key on the keypad, increase the speed of the stepper motor by keeping any higher number key on the keypad)

Appendix A

Keyboard Interface

※ Position Code

Key	0	1	2	3	4	5	6	7
Code	00	01	02	03	04	05	06	07
Key	8	9	A	B	C	D	E	F
Code	08	09	0A	0B	0C	0D	0E	0F
Key	:	STP	GO	REG	−	+	DA	AD
Code	10	11	12	13	14	15	16	17

※ Key Input Flowchart



Bibliography

Tech., MEDAS. 2008. *MDA 8086 Kit User Manual.* Korea : s.n., 2008.